DOCUMENT RESUME

ED 057 559 EM 009 395

TITLE	Project Solo; Newsletter Number Fourteen.
INSTITUTION	Pittsburgh Univ., Pa. Dept. of Computer Science.
SPONS AGENCY	National Science Foundation, Washington, D.C.
PUB DATE	17 Mar 71
NOTE	20p.; See also ED 053 566
EDRS PRICE	MF-$0.65 HC-$3.29
DESCRIPTORS	*Computer Assisted Instruction; Computer Programs; *Computer Science Education; Mathematics Instruction; *Programing; *Trigonometry
IDENTIFIERS	*Project Solo

ABSTRACT

Revised material from earlier Project Solo newsletters is presented here. The revised material updates programs to explain the law of sines and cosines and to apply the idea of rectangular coordinates approach to aircraft navigation systems such as VORTAC. A brief discussion of the value of off-line as opposed to on-line activity is also presented. (JY)

PROJECT SOLO

AN EXPERIMENT IN REGIONAL COMPUTING FOR SECONDARY SCHOOL SYSTEMS

University of Pittsburgh • Department of Computer Science • Pittsburgh, Pennsylvania 15213

Newsletter No. 14 March 17, 1971

Updates of Newsletters No. 1 and No. 7

The number of requests to receive the Project Solo newsletter is growing more rapidly than any of us had anticipated. Since our present budget does not permit reprinting, we will in general only be able to send more recent issues in response to new requests. To partly fill the gap, we have reprinted revised versions of the material that appeared in newsletters No. 1 and No. 7. Since new material has been added, these revisions are being sent with this newsletter (No. 14) to all persons on our mailing list. Example No. 3 on page 16 of "Some Principles for the Human Use of Computers in Education" might give teachers and students some new ideas about a less structured form of CAI. The best way to create files of the kind shown there is through QED.

Example:

```
-QED
*APPEND
SLIPPERY SOCK MENU:  }
HASH                 }  Everything typed here will be
OLD HASH             }  stored on the file /MENU/
VERY OLD HASH        }
D^C
*WRITE /MENU/
NEW FILE? YES
*EXIT
-DEF /MENU/ AS PUB
```

Off-Line Activity, Chinese Proverbs, and Piaget

A valuable aspect of solo mode computing is the off-line planning and analysis it invites, especially when these activities take place in concert with teachers and fellow students. Off-line activity is most effective when coordinated with on-line activity. This can be done for even simple topics, as the example /SOLID/ on the next page illustrates.

While using the computer (and plotter) on-line, the student approaches the learning ideal suggested in the Chinese proverb "I hear, and I forget; I see, and I remember; I do, and I understand". Off-line discussion can play the important role of recognizing what the student has done, thus establishing a fundamental union between the cognitive and affective domains. As Piaget writes, "social life affects intelligence...it provides (the student) a ready-made system of signs which modify his thoughts". The responsibility of teachers to give encouraging signs to students who have accepted and mastered the on-line challenge is large, and deserving of considerable attention.

---

*Supported in part by NSF grant GJ-1077

SAMPLE APPROACH TO MIXING OFF-LINE WITH ON-LINE ACTIVITIES:
Subject: Introduction to Cartesian Graphs.



Off-Line (Planning)

1. Explain use of rectangular coordinates in class - be brief, going immediately into...
2. Use of coordinate pairs by each student to describe a planar figure of his or her choice, to then be used...

On-Line (Active Use of Concepts)

```
>RUN 166TD /SOLID/

THIS PROGRAM WILL TRY TO PLOT A "3-D" PICTURE BASED ON A PLANAR
POLYGON YOU DESCRIBE.  PLEASE INPUT 2 COORDINATES (X,Y) AFTER EACH
"?" TO DESCRIBE THE VERTICES OF YOUR POLYGON.  START WITH THE VER-
TEX THAT HAS THE SMALLEST Y VALUE, AND THEN PROCEED COUNTER-
CLOCKWISE AROUND THE POLYGON.
FIRST...HOW MANY VERTICES DO YOU WISH
?5
NOW ENTER THE COORDINATES OF THE
VERTICES, WITH 1≤X≤14 AND 1≤Y≤9
?6,6
?7,6.5
?6.5,8
?5.5,7
?6,6.5
PLTL
```



(This then produces the plot shown above. This would normally be done on coordinate paper allowing the students to identify points.)

Off-Line (Analysis and Challenge)

3. What are the coordinates of the new points the plotter added to the points supplied by the student?
4. Can the student theorize what "rule" the plotter used to get the new points?
5. Go back to 2, trying to describe a more intricate figure that will amaze plotter, teacher, and friends!
6. Curious students will want (and should receive) the module that explains the plotter program.



GENERAL DISCUSSION FOR TEACHERS

Included in this section are listings of two programs which are working versions of the program assigned in the module, together with comments explaining them in detail. The first of these does not include the optional segment, and the Law of Sines and Cosines approach is taken (see problems 1 and 3 in the module). The second one includes the optional segment, and applies the conversion to rectangular coordinates approach (see problem 2 in the module).

The basic idea in the first program is to find both the angle at the aircraft, A3 (in the triangle formed by the aircraft, VORTAC and destination) and the direction of the VORTAC from the aircraft, Y. Then the direction of the destination from the aircraft, X, is computed as either Y + A3 or Y - A3. The decision regarding which of these is appropriate is as follows:

The number Z, is computed as the bearing of the destination from the VORTAC if the coordinate system is rotated so that the aircraft has a bearing of 0.* If Z is between 0 and π, then X is set equal to Y - A3; otherwise, it is set equal to Y + A3.

In the second program, the rectangular coordinates of the destination and plane are found, and the slope of the line between them is computed. The arctangent of that slope is then added to, or subtracted from π or 2π (= 0), as appropriate.

In the experience of the developers of the module, the rectangular coordinates approach is more straightforward and easier to program.

FOOTNOTE ON THE FOOTNOTE (page 3): Some students may be interested in the following quote from Norbert Wiener ("God and Golem")

> No, the future offers very little hope for those who expect that our new mechanical slaves will offer us a world in which we may rest from thinking. Help us they may, but at the cost of supreme demands upon our honesty and our intelligence. The world of the future will be an ever more demanding struggle against the limitations of our intelligence, not a comfortable hammock in which we can lie down to be waited upon by our robot slaves.

*The bearing of a point from the VORTAC will always mean the angular polar coordinate of that point, using the VORTAC as origin.

COMPLETED PROGRAM (SEGMENT 1)--TEACHER

```
1.11     demand E.
1.12     demand D.
1.13     demand S3.
1.14     demand S1.
```

Input statements for plane's bearing(E); destination's bearing (D); distance of destination from VORTAC (S1); and distance of plane from VORTAC (S3), respectively.

```
1.15     set D=D/360*2*3.14159.
1.16     set E=E/360*2*3.14159.
```

Conversion of bearings from degrees to radians

```
1.17     if S1 $ne 0, to step 1.21.
1.18     set S2=S3.
1.19     set X=D.
1.2      to step 1.85.
```

If the plane is at the VORTAC, the course is set equal to the bearing of the destination, and the distance of the plane from the destination is set equal to the distance of the destination from the VORTAC.

```
1.21     set A2=|D-E|.
1.215    if A2>3.14159,
                    set A2=6.28318-A2.
```

The angle A2 at the VORTAC (in the triangle determined by the plane, the VORTAC, and the destination) is found by taking the absolute value of the difference between the bearings of the plane and the destination. If this difference is greater than π, the angle is set equal to 2π minus the difference.

```
1.22     if |A2|>0.01, to step 1.224.
1.221    set S2=|S1-S3|.
1.222    if S1>S3, to step 1.2223.
1.2221   set X=D.
1.2222   to step 1.85.
1.2223   if E<3.14159,
    set X=E+3.14159; set X=E-3.14159.
1.223    to step 1.85.
```

If the plane and the destination have the same bearing, or close to it, the distance between the plane and the destination is set equal to the absolute value of the difference between the distances of the plane and destination from the VOR TAC (again, the magnitude is what's important). If the plane is closer to the VORTAC than the destination, the course is set equal to the bearing of the destination; otherwise the course is set equal to the bearing of the destination ± 180°.

```
1.224    if |A2-3.14159|>0.01,
                         to step 1.26.
1.23     S2=S1+S3.
1.24     set X=D.
1.25     to step 1.85.
```

If the plane, the destination, and the VORTAC lie in a line, or nearly so, with the plane and destination on opposite sides of the VORTAC, the course is set equal to the bearing of the destination. The distance between the plane and destination is set equal to the sum of the distance from the plane to the VORTAC and the distance from the VORTAC to the destination.

```
1.26    set S2=sqrt of
 (S3*S3+S1*S1-2*S3*S1*cosine of A2).
```

Application of the Law of Cosines to find the distance(S2) from the plane to the destination.

```
1.27    set S=(S3*sine of A2)/S2.
```

Application of the Law of Sines to find the sine of the angle(A3) at the plane in the triangle formed by the plane, the VORTAC, and the destination.

```
1.28    if |S| $ge 1, set C=0;
               set C=sqrt of (1-S*S).
```

Computation of the cosine of the angle(A3) at the plane. The positive value of the square root is used (see below).

```
1.29    if |C|<0.0001,
           set A3=1.570795;
        set A3=arc tangent of (S/C).
```

If the cosine is equal to or close to zero, the angle(A3) is set equal to π/2; otherwise, it is set equal to the arctangent of the sine over the cosine as computed above. The arctangent function in PIL returns for positive arguments values which are between 0 and π/2. Since the sine and cosine as computed above are positive, in this program the <u>acute</u> angle determined by the line from the VORTAC to the plane and the line from the destination to the plane is computed.

```
1.3     if S3*S3>S1*S1+S2*S2,
                  set A3=3.14159-A3.
```

When the obtuse angle at the plane is actually of interest, the supplement of the angle as computed above is found.

```
1.31    if E<3.14159,
   set Y=E+3.14159; set Y=E-3.14159.
```

The direction from the plane to the VORTAC (Y) is found. This is the opposite (±180°) to the bearing of the plane from the VORTAC.

```
1.33    set Z=D-E.
1.34    if Z<0,set Z=Z+6.28318.
1.35    if Z<3.14159,
          set X=Y-A3;set X=Y+A3.
```

The bearing of the destination(Z), with respect to a line from the VOR TAC to the aircraft, is computed. If Z is less than π, A3 is subtracted from Y to find the course; otherwise it is added to Y.

```
1.85    set XX=X*360/(2*3.14159).
```

The course from the present location of the plane to the destination is converted from radians to degrees.

```
1.855   if XX<0, set XX=XX+360.
1.856   if XX>360, set XX=XX-360.
1.86    type "direction and
          distance of destination.".
1.87    type XX,S2.
1.99    done.
```

The course is converted, if necessary, to an equivalent value between 0° and 360°. The course and distance to the destination are printed.

```
1.13    type "Enter distance of plane from VORTAC.".
1.14    demand S1.
1.15    type "Enter direction of plane from VORTAC,".
1.16    type "in degrees clockwise from north.".
1.17    demand E.
1.23    type "Enter distance of destination from VORTAC.".
1.24    demand S3.
1.25    type "Enter direction of destination from VORTAC".
1.26    type "in degrees clockwise from north.".
1.27    demand D.
```

Input section.

```
1.32    set D=D/360*2*3.14159.
1.33    set E=E/360*2*3.14159.
```

Conversion from degrees to radians.

```
1.35    if S1 $ne 0, to step 1.4.
1.36    set S2=S3.
1.37    set X=D.
1.38    to step 1.85.

1.4     do part 3.
```

If the plane is at the VORTAC, the course is set equal to the bearing of the destination, and the distance of the plane from the destination is set equal to the distance of the VORTAC from the destination.

```
1.85    set XX=X*360/(2*3.14159).
```

Conversion of direction to the destination from radians to degrees.

```
1.855   if XX < 0, set XX=XX+360.
```

If the direction to the destination is negative, the equivalent positive angle is found.

```
1.86    type "Direction and distance of destination".
1.87    type XX,S2.
```

Prints out the direction to and distance from the destination.

```
1.9     type "If you wish to fly further,type yes in quotes.".
1.91    demand answer.
1.96    if answer="yes",do part 2;done.

1.97    to step 1.35.
```

Person running program is questioned regarding whether he wants to see the effect of the wind after some elapsed time. If he answers 'yes,' part 2 is executed; otherwise, program is finished.

After execution of part 2, control is transferred to step 1.35.

```
2.0     type "Enter plane's speed."
2.01    demand V1.
2.021   type "Enter wind's speed.".
2.022   demand W1.
2.023   Type "Enter direction fro  which wind is blowing,".
2.024   type "in degrees clockwise from north.".
2.025   demand D1.
```

Input section for part 2. Speeds are in miles per hour.

```
2.04    if D1 $ge 180,set D1=D1-180;set D1=D1+180.
```

The direction from which the wind is blowing is converted to the direction to which it is blowing.

```
2.041   set D1=D1/360*2*3.14159.
```

The wind's direction is converted to radians.

```
2.05    type "Enter an elapsed time
                            in minutes.".
2.051   demand T.
```

Elapsed time is input.

```
2.052   set T=T/60.
```

Time is converted from minutes to hours.

```
2.053   type "Enter heading.".
2.054   demand H.
2.055   set H=H/360*2*3.14159.
```

Heading is input and converted to radians.

```
2.06    set N1=
        (V1*sine of H + W1*sine of D1)
                    *T +S1*Sine of E.
2.07    set N2=
        (V1*cos of H + W1*cos of D1)
                    *T +S1*Cosine of E.
```

New rectangular coordinates of the plane after it has flown for the elapsed time with the heading input above are computed. The effect of the wind is taken into account. N1 is the east-west coordinate and N2 is the north-south coordinate.

```
2.08    set S1=Sqrt of (N1*N1+N2*N2).
```

New distance of the plane from the VORTAC is calculated.

```
2.11    if |N2|>0.0001,to step 2.21.
2.12    if N1 >0, set E=1.570795.
2.13    if N1 <0, set E=4.712388.
2.14    done.
```

If the plane has a zero (or very small) north-south coordinate, the bearing of the plane is set equal to: (1) $\pi/2$ if the plane is east of the VORTAC; (2) $3\pi/2$ if the plane is west of the VORTAC.

```
2.21    set A4=Arc tangent of (|N1/N2|).
```

The angle(A4) between a line from the VORTAC to the plane and the east-west axis is found by computing the arctangent of the ratio of the absolute values of the coordinates.

```
2.22    if N1 <0, to step 2.31.
2.23    if N2>0, set E=A4;
                    set E=3.14159-A4.
2.24    done.
2.31    if N2<0, set E=3.14159+A4;
2.32    done.       set E=6.28318-A4.
```

The angle(A4) is either added to or subtracted from 0 (= 2π) or π depending on which quadrant the plane is in to find the plane's new bearing.

```
3.01    set F1=S1*sin of E.
3.02    set F2=S1*cos of E.
3.03    set B1=S3*sin of D.
3.04    set B2=S3*cos of D.
```

The rectangular coordinates of the plane and the destination are computed from the corresponding polar coordinates.

```
3.1     for N1=F1-B1:set N2=F2-B2.
```

The difference between the east-west coordinates N1 of the plane and destination and the corresponding difference between the north-south coordinates N2 are computed.

```
3.11    set S2=sqrt of (N1*N1+N2*N2).
```

The Pythagorean theorem is used to calculate the distance from the plane to the destination.

```
3.12    if S2 $ge 4,to step 3.21.
3.13    type "You are within sight
                     of destination.".
3.2     done.
```

If the destination is less than four miles from the plane, the program types a message and stops.

```
3.21    if |N2|>0.001, to step 3.31,
3.22    if N1>0,set X=4.712388.
3.23    if N1<0,set X=1.570795.
3.24    done.
```

If there is no difference (or nearly so) in the north-south coordinates of the plane and destination the course X is set equal to:

(1) $3\pi/2$ if the destination is west of the plane

(2) $\pi/2$ if the destination is east of the plane.

```
3.31    set A4=arc tangent of (|N1/N2|).
```

The angle(A4) between the line from the plane to the destination and the east-wast axis is computed.

```
3.32    if N1<0,to step 3.41.
3.33    if N2>0,set X=3.14159+A4;
                      set X=6.28318-A4.
3.34    done.
3.41    if N2<0,set X=A4;
                      set X=3.14159-A4.
3.42    done.
```

The angle(A4) is either subtracted from or added to 0 ($= 2\pi$) or π depending on the relative positions of the plane and destination, to determine the course X to the destination from the plane.

COMPLETED PROGRAM (SEGMENT 1)--TEACHER

```
11100 INPUT E
11200 INPUT D
11300 INPUT S3
11400 INPUT S1
```

Input statements for plane's bearing (E); destination's bearing (D); distance of destination from VORTAC (S3); and distance of plane from VORTAC (S1), respectively.

```
11500 LET D=D/360*2*3.14159
11600 LET E=E/360*2*3.14159
```

Conversion of bearings from degrees to radians

```
11700 IF S1 <> 0 GØTØ 12100
11800 LET S2=S3
11900 LET X=D
12000 GØTØ 18500
```

If the plane is at the VORTAC, the course is set equal to the bearing of the destination, and the distance of the plane from the destination is set equal to the distance of the destination from the VORTAC.

```
12100 LET A2=ABS(D-E)
12150 IF A2<=3.14159 GØTØ 12200
12160 LET A2=6.28318-A2
```

The angle A2 at the VORTAC (in the triangle determined by the plane, the VORTAC, and the destination) is found by taking the absolute value of the difference between the bearings of the plane and the destination. If this difference is greater than π, the angle is set equal to 2π minus the difference.

```
12200 IF ABS(A2)>0.01 GØTØ 12240
12210 LET S2=ABS(S1-S3)
12220 IF S1>S3 GØTØ 12223
12221 LET X=D
12222 GØTØ 18500
12223 IF E<3.14159 GØTØ 12226
12224 LET X=E-3.14159
12225 GØTØ 18500
12226 LET X=E+3.14159
12230 GØTØ 18500
```

If the plane and the destination have the same bearing, or close to it, the distance between the plane and the destination is set equal to the absolute value of the difference between the distances of the plane and destination from the VOR TAC (again, the magnitude is what's important). In this case, the course is set equal to the bearing of the destination when the plane is farther from the VORTAC than the destination is.

```
12240 IF ABS(A2-3.14159)>0.01
                         GØTØ 12600
12300 LET S2=S1+S3
12400 LET X=D
12500 GØTØ 18500
```

If the plane, the destination, and the VORTAC lie in a line, or nearly so, with the plane and destination on opposite sides of the VOR TAC, the course is set equal to the bearing of the destination. The distance between the plane and destination is set equal to the sum of the distance from the plane to the VORTAC and the distance from the VORTAC to the destination.

```
12600 LET S2=SQR
      (S3*S3+S1*S1-2*S3*S1*COS(A2))
```

Application of the Law of Cosines to find the distance(S2) from the plane to the destination.

```
12700 LET S=(S3*SIN(A2))/S2
```

Application of the Law of Sines to find the sine of the angle(A3) at the plane in the triangle formed by the plane, the VORTAC, and the destination.

```
12800 IF ABS(S)>=1 GOTO 12850
12820 LET C=SQR(1-S*S)
12825 GOTO 12900
12850 LET C=0
```

Computation of the cosine of the angle(A3) at the plane. The positive value of the square root is used (see below).

```
12900 IF ABS(C)<0.0001 GOTO 12950
12920 LET A3=ATN(S/C)
12925 GOTO 13000
12950 LET A3=1.570795
```

If the cosine is equal to or close to zero, the angle(A3) is set equal to $\pi/2$; otherwise, it is set equal to the arctangent of the sine over the cosine as computed above. The arctangent function in BASIC returns for positive arguments values which are between 0 and $\pi/2$. Since the sine and cosine as computed above are positive, in this program the acute angle determined by the line from the VORTAC to the plane and the line from the destination to the plane is computed.

```
13000 IF S3*S3<=S1*S1+S2*S2
                        GOTO 13200
13100 LET A3=3.14159-A3
```

When the obtuse angle at the plane is actually of interest, the supplement of the angle as computed above is found.

```
13200 IF E<3.14159 GOTO 13250
13220 LET Y=E-3.14159
13225 GOTO 13300
13250 LET Y=E+3.14159
```

The direction from the plane to the VORTAC (Y) is found. This is the opposite (±180°) to the bearing of the plane from the VORTAC.

```
13300 LET Z=D-E
13400 IF Z>=0GOTO 13500
13450 LET Z=Z+6.28318
13500 IF Z<3.14159 GOTO 13550
13520 LET X=Y+A3
13525 GOTO 18500
13550 LET X=Y-A3
```

The bearing of the destination(Z), with respect to a line from the VOR TAC to the aircraft, is computed. If Z is less than π, A3 is subtracted from Y to find the course; otherwise it is added to Y.

```
18500 LET X1=X*360/(2*3.14159)
```

The course from the present location of the plane to the destination is converted from radians to degrees.

```
18550 IF X1>=0 GOTO 18560
18555 LET X1=X1+360
18560 IF X1<=360 GOTO 18600
18565 LET X1=X1-360
18600 PRINT "DIRECTION AND
            DISTANCE OF DESTINATION."
18700 PRINT X1,S2
19900 END
```

The course is converted, if necessary, to an equivalent value between 0° and 360°. The course and distance to the destination are printed.


A1
S3
S2
S1
A3
Y
X



Phantom VORTAC



● This module will guide you in preparing the master program for an on-board flight computer. The computer takes information from a VHF omni range system (VORTAC), and calculates the magnetic course and distance to a given airport for the pilot. The output of the program is such that the pilot can fly toward a "phantom" VORTAC located at any airport he selects.

● A description of this newly developed navigational system and the mathematics on which it is based are given on page 2.

● Pages 3 and 4 outline alternate methods for handling the computation, and suggest ways in which previous programs you have written might be incorporated as sub-routines.

● A "real time" simulation of a flight using this system is suggested as an advanced level program.

Pilots flying over the United States (and most other countries of the world) rely on radio facilities called VORTACs* for navigational information. The basic information the pilot receives in the cockpit is his position relative to the VORTAC, given in polar coordinates.

The pilot in the illustration below would describe his position (obtained from his radio instruments) as being "on the 100° radial of the Allegheny County (AGC) VOR, 50 miles out".

It is easy for this pilot to note that he can get to AGC by turning right, and flying a course** of 280° (Why 280°?). If he is going 200 miles per hour, it is also easy for him to estimate that he will arrive at AGC in 15 minutes (Is this exactly true? --go back over the module on vector addition if you are not sure.)



The catch to all of this is that the location of the VOR is usually different from the location of the destination airport. This difficulty can be handled by flying two legs, the first from the present position to the VOR, and the second from the VOR to the airport. This is obviously an inefficient route.



A new navigational system uses an on-board, special-purpose computer to tell the pilot what course and distance to fly in order to go directly from his present position to the airport. Let's first examine three ways of analyzing the mathematics involved in such a computation.

---

*Very high frequency Omni Range and TACAN, where TACAN is an older military system of distance measuring equipment. Most civilian pilots call these facilities VOR-DME stations.

**The angular direction of the intended flight path, measured clockwise from N.

Before writing a program for such an on-board computer, it would be useful for a programmer to solve by hand typical problems arising in this situation.* At this point, let's look at three such solutions.



Problem 1. The aircraft is 150 miles out on the 135° radial and the destination airport is 100 miles out on the 210° radial. That is, in the diagram, S1 = 150, E = 135°, S3 = 100, D = 210°.

Let A2 be the angle determined by S1 and S3. Hence, A2 = 210° - 135° = 75°. By the Law of Cosines,

$$S2 = \sqrt{100^2 + 150^2 - 2 \cdot 100 \cdot 150 \cdot \cos(75°)} = 157.3 \text{ miles.}$$

Now by the Law of Sines, $\sin A3 = (100 \cdot .966)/157.3 = .614$ and $\cos A3 = \sqrt{1 - .614^2} = .789$. From this we can get A3 = arctan(.614/.789) = 38°. (Why wasn't A3 computed directly from sin A3?) Therefore X = 135° + 180° - 38° = 277°. Output to the pilot is 157.3 miles, 277°--is this answer reasonable?

Problem 2. The airplane is 250 miles out on the 45° radial and the destination airport is 100 miles out on the 280° radial. In terms of the diagram, S1 = 250, E = 45°, S3 = 100, D = 280°.

Converting the positions of the aircraft and destination to rectangular coordinates with the VORTAC as origin we have

$x_p = 250 \cdot \sin 45° = 176.8$ miles,
$y_p = 250 \cdot \cos 45° = 176.8$ miles,
$x_d = 100 \cdot \sin 280° = -98.48$ miles,
$y_d = 100 \cdot \cos 280° = 17.36$ miles.

$$A = \arctan \frac{176.8 - 17.36}{176.8 - (-98.48)} = \arctan(.5792) = 30.1°$$



*This is another way of saying that computers do not remove the responsibility of analyzing a problem before solving it--quite the contrary; they demand more thought than ever. This may be, in fact, one of the most important contributions computing systems can make to learning.

Therefore $X = 270° - 30.1° = 239.9°$. By the Pythagorean Theorem $S2 = \sqrt{(176.8 - 17.36)^2 + (176.8 - (-98.48))^2} = 318.1$. Hence the output to the pilot is: 318.1 miles, 239.9°.



Problem 3. The aircraft is 150 miles out on the 120° radial and the destination airport is 400 miles out on the 150° radial.** In the diagram, E = 120°, S1 = 150 miles, D = 150°, S3 = 400 miles. Using the same reasoning as in Problem 1 yields:

$A2 = 30°$

$S2 = \sqrt{400^2 + 150^2 - 2 \cdot 150 \cdot 400 \cdot \cos 30°}$

$= 280.3$ miles

$\sin A3 = 400 \cdot (.5236/280.3) = .7135$

$\cos A3 = \sqrt{1 - .7135^2} = .7007$

$A3 = \arctan(.7135/.7007) = 45.52°$

$X = 120° + 180° - 45.52° = 254.48°$

Output to the pilot is: 280.3 miles, 254.48°--does this answer look reasonable? What went wrong?

Assignment. Write a program for an on-board navigational computer which will accept as input the radial and distance from a VORTAC of both an aircraft and a destination, and which will compute a course and distance to the destination. The trigonometric subroutines in the computer require arguments in radians, but pilots think in terms of degrees, so it will be necessary for you to convert degrees to radians and back (see the module on converting to radians). Other possibly useful topics are inverse trigonometric functions*, the Laws of Sines and Cosines*, and transformation of polar to rectangular coordinates*.

---

*See modules you have worked previously.

**By now you should have noticed that the word radial is used to designate the angular position of a line segment that starts at the VORTAC.

Optional Section. During most flights, the aircraft moves through the air and the air moves as well. Write an addition to your program which will:

1. Accept as additional input, specified by the 'pilot', (a) the aircraft's speed, (b) the speed of the wind, (c) the direction of the wind, (d) an elapsed time, t, and (e) a heading* for the aircraft. Any heading should be acceptable: the 'pilot' should be able to fly wherever he likes.

2. Compute the new position of the aircraft on the basis of the above information.

3. Repeat steps 1 and 2 as often as desired or necessary to reach the destination airport.

You can look at the input/output sequence of one program that accomplishes this and has been written and stored.

In order to gain access to this program, please logon (directions are given in an introductory module) using the User Identification Code given to you by your teacher:

In order to interact with this stored program type in the statements:

From this point on the operation of the program should be self-explanatory.

*Heading is defined as the angular direction of the longitudinal axis of the aircraft with respect to North. In the picture below, the pilot is flying a course of 90°, but his heading is 65°. Why?



GROSS FLOW CHART FOR SUGGESTED ASSIGNMENTS



AIR RACE GAME

1. Logon the computer and type:
   -EXE 159TD /VOR/

2. Each player types in a starting location for his plane, and the location of his destination airport. The players should also mark these locations on the chart at the right as shown in the sample. All locations are with respect to the VOR radio station located at the origin. Angular directions are given in degrees measured clockwise from North, distances in miles.

3. The speed of the wind and the direction (in degrees) from which it is blowing are entered. Example: A 20 mph wind which blows from West to East has a direction of 270, and a speed of 20.

4. Each player then enters a speed, a heading, and an elapsed time for his plane. "Heading" is the direction (measured clockwise from North) in which you wish to point your plane. Example: If you wish to point your plane Southwest, use a heading of 225. The object is to reach your destination airport.

5. The computer will now type out a new location, based on the data you just gave it. Repeat step 4 until winning player sees the message "You are within sight of destination (3 miles)."

*See attached interaction and the module "Phantom VORTAC" for further details.



SAMPLE INTERACTION WITH THE AIR RACE GAME

(LOGON)

-EXE 159TD /VOR/

```
ALL DIRECTIONS ARE MEASURED IN DEGREES
CLOCKWISE FROM NORTH
ENTER DISTANCES OF PLANES FROM VORTAC
PLANE 1
?380
PLANE 2
?440
ENTER DIRECTIONS OF PLANES FROM VORTAC
PLANE 1
?50
PLANE 2
?320
ENTER DISTANCES OF DESTINATIONS FROM VORTAC
PLANE 1
?350
PLANE 2
?280
ENTER DIRECTIONS OF DESTINATIONS FROM VORTAC
PLANE 1
?200
PLANE 2
?130
ENTER WIND DIRECTION
?180
ENTER WIND SPEED
?10
PLANE 1
POSITION RELATIVE TO VORTAC
DISTANCE=    380  MILES
RADIAL=   50   DEGREES
COURSE AND DISTANCE TO DESTINATION
 215.6308206   DEGREES          705.1683936   MILES
PLANE 2
POSITION RELATIVE TO VORTAC
DISTANCE=    440  MILES
RADIAL=   320  DEGREES
COURSE AND DISTANCE TO DESTINATION
 136.1135549   DEGREES          717.3958114   MILES
IF YOU WISH TO FLY FURTHER TYPE YES.
?YES
ENTER PLANES' SPEEDS.
PLANE 1
?115
PLANE 2
?110
ENTER HEADINGS.
PLANE 1
?210
PLANE 2
?140
```

```
ENTER ELAPSED TIMES.
PLANE 1
?360
PLANE 2
?380
pLANE 1
POSITION RELATIVE TO VORTAC
DISTANCE=     298.2108681    MILES
RADIAL=   190.4134233    DEGREES
COURSE AND DISTANCE TO DESTINATION
 241.5915096    DEGREES             74.81478979    MILES
PLANE 2
POSITION RELATIVE TO VORTAC
DISTANCE=     212.094038     MILES
RADIAL=   128.9339583    DEGREES
COURSE AND DISTANCE TO DESTINATION
 133.3237435    DEGREES             68.05716319    MILES
IF YOU WISH TO FLY FURTHER TYPE YES.
?YES
ENTER PLANES' SPEEDS.
PLANE 1
?115
PLANE 2
?110
ENTER HEADINGS.
PLANE 1
?245
PLANE 2
?135
ENTER ELAPSED TIMES.
PLANE 1
?40
PLANE 2
?40
PLANE 1
POSITION RELATIVE TO VORTAC
DISTANCE=     342.0614764    MILES
RADIAL=   201.1436708    DEGREES
COURSE AND DISTANCE TO DESTINATION
 159.5452339    DEGREES             10.5223412    MILES
PLANE 2
POSITION RELATIVE TO VORTAC
DISTANCE=     280.8389771    MILES
RADIAL=   129.4568428    DEGREES
COURSE AND DISTANCE TO DESTINATION
 237.2440248    DEGREES             2.78757797    MILES
YOU ARE WITHIN SIGHT OF DESTINATION.
IF YOU WISH TO FLY FURTHER TYPE YES.
?NO
```